
Rascal is a meta programming language for processing source code in the broad sense (models, doc-
uments, formats, languages, etc.). In this short note we discuss the implementation of the ‘TTC’14
FIXML to Java, C# and C++ Case” in Rascal. In particular we will highlight the use of string tem-
plates for code generation and relational analysis to deal with dependency-based ordering problems.

1 Introduction

Rascal is a meta programming language for source code analysis and transformation [1, 2]. Concretely,
it is targeted at analyzing and processing any kind of “source code in the broad sense”; this includes
importing, analyzing, transforming, visualizing and generating, models, data files, program code, docu-
mentation etc.

Although Rascal features a Java-like syntax, it is a functional programming in that all data is im-
mutable (implemented using persistent data structures), and function programming concepts are used
throughout: algebraic data types, pattern matching, higher-order functions, comprehensions etc.

Specifically for the domain of source code manipulation, however, Rascal features powerful primi-
tives for parsing (context-free grammars), traversal (visit statement), relational analysis (transitive clo-
sure, image etc.), and code generation (string templates). The standard library includes programming
language grammars (e.g., Java), IDE integration with Eclipse, numerous importers (e.g. XML, CSV,
YAML, JSON etc.) and a rich visualization framework.

In the following sections we discuss the realization of the TTC’14 tasks in Rascal. We conclude the
paper with some observations and concluding remarks. All code examples can be found online at:

https://github.com/cwi-swat/ttc2014-fixml-case

2 The transformation

As proposed in the description of the case study, the solution transformation has been broken down into
the following sub transformations:

1. XML text to model of XML metamodel

2. model of XML metamodel to a metamodel of OO programming languages

3. OO metamodel to program text (for different OO programming languages)

The first item is readily addressed by Rascal’s standard library, which contains a meta model for XML
and (de)serialization functions. Below we discuss the remaining tasks.

2.1 Sub transformation 2: XML metamodel to OO metamodel

We have defined an OO metamodel that specifically for the purpose of this particular task. This means
that it is not comprehensive enough to model arbitrarily complex OO system, but it serves as an interme-
diate model from which all the desired output in the context of this task can be generated. For instance,
it is possible to see that the types for representing fields possess an altName field. This is needed to rep-
resent unambiguous parameters in the case of C++ code, as required by the particular code style shown
in the description of the use case.

The following datatypes capture the structure of the OO metamodel:

data OOModel = oomodel(list[Class] classes);

data Class = class(str name, list[Field] literalFields, list[Field] objFields);

data Field = objField(Type tipe, str name, str altName, list[Field] vals)

| literalField(Type tipe, str name, str altName, str val);

data Type = tipe(str className);

An OOModel consists of a list of classes. Each class has a name, a number of literal fields, and a
number of object fields. A field can be either a object field or a literal field. The difference is that object
fields can have sub fields, whereas literal fields are directly initialized with a (primitive) value. Types
are represented by the Type data type. Note that the difference between literal fields and object fields
is encoded in the Field type; however, for convenience, the class constructor also distinguishes them
explicitly.

In order to map a FIXML model to a OO model, it was necessary to bridge the conceptual gap
between both by specifying a transformation from a generic XML model to an OO class using the inter-
pretation described in the description of the case study. This transformation was specified in 75 SLOC
of Rascal code.

2.2 OO model to program text

Once we have the OO model, the final step is to serialize it as a program in three different OO languages:
Java, C#, and C++. The three transformations are quite similar. The main differences are related to
particular idioms of one implementation in respect to the others, particularly in the case of C++. For
instance, although the order in which classes are declared is not relevant in the case of Java and C#, it
matters in the case C++, given its declare-before-use policy. For this reason, we just present the source
code of the Java serialization, and discuss afterwards how we addressed the specific particularity of the
C++ transformation.

str class2javaClass(class(name, literalFields, objFields)) =

"class <name> {

’ <fields2javaFields(literalFields, objFields)>

’ <name>(){ }

’ <fields2constructor(name, literalFields, objFields)>

’}";

str fields2constructor(str className, list[Field] literalFields, list[Field] objFields)=

"<className>(<toParameters(literalFields, objFields)>){

’ <for (literalField(_, name, _, _) <- literalFields){>

’ this.<name> = <name>;

’ <}>

’ <for (objField(_, name, altName, _) <- objFields){>

’ this.<name> = <altName>;

’ <}>

’}";

str fields2javaFields(list[Field] litFields, list[Field] objFields) =

"<for (literalField(tipe, name, _, val) <- litFields){>

’ <tipe.className> <name> = \"<val>\";

’<}>

’<for (objField(tipe, name, _, vals) <- objFields){>

’ <tipe.className> <name> = new <tipe.className>(<toArguments(vals)>);

’<}>";

The three functions produce strings using Rascal’s string templates. These templates support multi-
line strings (margins indicated by ’), string interpolation (escaping expressions with the < and > charac-
ters) and automatic indentation. As a result, model-to-text transformations are very easy to express.

As mentioned before, the declare-before-use policy of C++ had to be taken into account. We solve
this problem by first sorting the list classes according to their dependencies (topological order):

list[Class] orderClasses(list[Class] classes) =

[classesMap[cName] | cName <- reverse(analysis::graphs::Graph::order(depGraph))]

when classesMap := (className: c | c:class(className, _, _) <- classes),

depGraph := {<className, oName> | class(className, _, oFields) <- classes

, objField(tipe(oName), _, _, _) <- oFields};

The orderClasses function uses the order function from the analysis::graphs::Graph module,
which computes the topological order of the nodes in a graph. Therefore, the only required task in order
to implement the declare-before-use policy was to create a dependency graph between the classes in
the model. The local variable depGraph receives its value from a comprehension that we will explain,
as a good example of the advantage of combining Rascal’s functional nature and its relational calculus
support. Given a set of classes, the comprehension builds a set of tuples (i.e., a binary relation) where its
first member is the name of one class being iterated in the first level of iteration, and the second member
corresponds to the class name of an object field of such a class. This value is obtained by adding another
level of iteration for the object fields of each class.

3 Concluding Remarks

Implementing the FIXML case study in Rascal was straightforward, as Rascal was indeed designed
for supporting the analysis and transformation of source code artifacts. Because of this, many of the
more complex tasks were already solved using the standard library, e.g., XML parsing and topological
sorting. In sum, it took only 179 SLOC to implement the pipeline required to output the code in the three
required languages. The most complex part of the assignment was to identify the minimal subset of an
OO metamodel that we needed in order to implement this particular use case. By doing that, we avoided
unnecessary accidental complexity and conceived a metamodel that was described in just 13 SLOC.

